CHAPTER 4

USING SPREADSHEETS TO COMPUTE STOCK AND OPTION TREES

The future is already here. It's just unevenly distributed.

William Gibson

4.1 SOME SPREADSHEET BASICS

In Chapter 3 we learned that each node in a stock tree (such as that shown in Figure 4.1) has a value, S_k, that is obtained by multiplying a neighboring value by one of two constants, u or d.

The actual process of filling in these values is tedious. A spreadsheet is a very useful tool for eliminating most of these steps. To see how this works, let us look at a typical blank spreadsheet page that appears on the computer screen when a spreadsheet program is opened (Figure 4.2). Each "cell" in this grid can hold either a title such as "Stock Price," a numerical value, or a formula.

As a first step, enter the value 100 in cell J1. This is done by selecting the cell with the mouse, typing **100**, and hitting the Enter (or Return) key. As you type, you should see 100 first appear on a special line above the cells. After you hit Enter, the number will appear in cell J1.



FIGURE 4.1
Stock tree

	A	B	C	D	E	F	G	H	I	J
1										
2										
3										
4										
5										
6										

FIGURE 4.2
Blank spreadsheet

Next, make a stock tree that branches down from J1. The next two nodes of the stock tree will be I2 and K2. To produce the entry for I2, select it with the mouse and type

=.9*J1

Although this **formula** appears on the special top line as you type, after you hit Enter the cell will show a numerical value, 90. The spreadsheet has retained this formula in cell I2.

This is a good point at which to pause. **Save** the spreadsheet you are creating, and give it a file name such as "Stock Tree" or "stock."

The formula we created involved the number $d = 0.9$. There is a simple way to make our stock tree more flexible so that it accepts other values of d. Let us agree that we will always enter the correct d value in cell A7. In fact, enter **.9** in A7 and enter the title **d value** in cell B7 to remind us where our d parameter is located. We wish to treat the initial stock value in the same manner. Enter 100 in **A1** and the title **Stock Price** in cell B1. To make the full titles visible, you should change the width of column **B**.

	A	B	C	D	E	F	G	H	I	J	K
1	100	Stock Price								100	
2									90		
3											
4											
5											
6											
7	.9	d value									

FIGURE 4.3
Spreadsheet with six entries

To use the entries in column A, just replace the entry in cell J1 by the simple formula, **=A1**. Also, replace the formula in cell I2 with **=J1*$A$7**. The numerical values will remain the same. The new feature is that if we change either the stock price or the d value, then the I and J columns change accordingly. Now your spreadsheet should look like Figure 4.3.

Why did we put dollar signs in the I2 formula? They signal the spreadsheet not to change the A7 address as we copy and paste this formula. The next step is to copy the formula by selecting the cell I2 and choosing a copy command. Then select the new cell, H3, and paste the formula there. Notice that the value there is 81. Continue to paste the *original formula* in cells G4, F5, and E6. These are locations down one row and left one column from the previous cell. If you check the formula entry for any of these cells, you will see that it multiplies the cell value of its upper right neighbor by the d value. So we have one edge of our stock tree filled, and this edge can accept any change of either the d value or the initial stock price.

We can fill in more of the stock tree if we use a u value. Let us agree that the u value will be entered into A6. In fact, enter the title **u value** in B6. It is good practice to **save** the spreadsheet file again since we have invested some effort into creating these cells and formulas. We suggest even saving a copy of this file with another file name, such as "Template". We will use these formulas again soon.

Now enter the u value, **1.1**, in A6. Next, select the cell that is down and to the *right* of the beginning node of our stock tree. This is K2. Type the formula **=J1*$A$6** and hit Enter. Your spreadsheet should look like Figure 4.4.

You will enjoy completing the final step. Select the formula we just created in K2 and choose a copy command. Next, paste the formula into any cell that is down and to the right of a location containing a numerical value. A good place to start is at J3, and you can immediately paste at the locations K4 and L5 to see the sequence 99, 109, and 120 appear. Continue to fill out the entire stock tree this way. When partially filled out, the tree should appear like Figure 4.5.

If you succeeded in producing these values, save this sheet once more (still as "Stock tree" or "Stock") and save another copy under a different file name as well. This sheet will be valuable for working exercises that include finding the prices of options, too.

	A	B	C	D	E	F	G	H	I	J	K
1	100	Stock Price								100	
2									90		110
3								81			
4							72.9				
5						65.6					
6	1.1	u value			59						
7	.9	d value									

FIGURE 4.4
Spreadsheet after *u* branch is begun

	A	B	C	D	E	F	G	H	I	J	K
1	100	Stock Price								100	
2									90		110
3								81		99	
4							72.9		89.1		109
5						65.6		80.2		98	
6	1.1	u value			59		72.2		88.2		
7	.9	d value									

FIGURE 4.5
Spreadsheet with tree partially filled by pasting

EXERCISES

1. Figure 4.1 has a stock tree with $u = 1.1$ and $d = 0.9$. Reproduce this stock tree with your spreadsheet and print the result.
2. Exercise 1 of Section 3.4 has a stock tree with $S_0 = 70$ and three time periods. Reproduce this stock tree with your spreadsheet and print the result.
3. Suppose that you edit your "Stock" spreadsheet in the following way. Instead of entering the d value into A7, you enter the formula **= 1/A6** in A7. Now choose $S_0 = 100$, $u = 1.2$ in this new spreadsheet and print the result of six time periods. Explain why several stock values are repeated.
4. Use the edited spreadsheet of Exercise 3 to print the stock tree of Exercise 1 of Section 3.4. Explain why you cannot reproduce Figure 4.1 with this method.

4.2 COMPUTING EUROPEAN OPTION TREES

Recall from Section 3.2 that we can use a stock tree to immediately write out an option price, such as a call, on its date of expiration. The spreadsheet we created in Section 4.1 will be updated to perform the same calculation for us. It will have the dynamic feature that as soon as we change the u or d value, the option price will change automatically. Of course, our ultimate goal is to fill in the entire option tree to obtain earlier prices.

As a first step, let us decide where to locate our option tree on our spreadsheet page. We want to use the *same* page we created for Figure 4.2 in order to use the stock values from that tree. We will locate the option tree between rows 11 and 20. Any option entry will be in the same column as the stock price it is based on. This means that the top option entry will be in J11.

The possible call prices on the date of expiration depend on the number of steps in the stock tree. Record the number of steps in A3 and enter the title **# Periods** in B3. Also, we need a location for the *strike price*, so enter the value **105** in A2 and the title **Strike Price** in B2.

The stock values after five periods are on line 6 of the spreadsheet. So we will fill in the option values on line 16 (putting the top of our tree at J11). What are these option values? They are 0 if the stock price is lower than the strike price, and they are the difference between the price and the strike if the stock is higher than the strike. The spreadsheet has a formula for this.

The formula for K16 is

=MAX(K6-$A$2,0)

Enter this formula into K16 (you do not need to capitalize letters). You should **save** this spreadsheet and give it a different file name such as "Euro Call." The formula you just entered refers to K6 to obtain the stock value, and subtracts the number in A2 from it. If the difference is negative, it records a 0 in the cell. The built in MAX function is extremely convenient and helpful.

Next, copy this formula and paste it in several locations on line 16, in columns E, G, I, M, and O. Line 16 of your spreadsheet should look as shown in Figure 4.6.

We would like to use the chaining method, introduced in Section 3.2, to fill in the top portion of the option tree. This procedure relies on two quantities that were calculated in Section 3.2. Soon we will devise formulas for the correct *arbitrage pricing probability*, q, and the discount factor, but for now we will simply enter these values as 0.7564 in cell A12 and 0.95 in cell A11.

	E	F	G	H	I	J	K	L	M	N	O
11											
12											
13											
14											
15											
16	0		0		0		2.81		26.8		56.1
17											

FIGURE 4.6
One row of option tree

You should also enter the titles **q** and **exp –r** in the adjoining cells, B12 and B11. For convenience, enter $1 - q$ in A13 by entering the formula **=1-$A$12** in that cell.

Now we can fill in J15 with the chaining value for the option. If you look at formula (2.6) in Section 2.3.3, you will see that we should average the cell entries in I16 and K16 and then discount this average. The formula we enter (in J15) is

=$A$11*($A$13*I16+$A$12*K16)

This formula is really

$$e^{-r}[\,(1-q)\text{I16} + q\text{K16}]$$

in disguise. The number 2.02 should appear in the cell.

We have to type this formula in *only once*. If we had to type it for every entry, we would probably reject spreadsheets as inefficient and unwieldy. We can copy this formula once and paste it at all the nodes of our option tree. If you do this, your topmost entry will be 19.7. This is the chaining value for the present price of the option.

The Refined Spreadsheet Although our option-stock tree spreadsheet is certainly worth **saving**, it suffers from the defect that the discount factor and the q value must be entered. Let us correct this defect. We will use A5 as a location to enter the interest rate; enter the corresponding title in B5.

To use this cell in our calculations, enter the title **exp r** in B10 and enter the following formula in A10:

= EXP($A$5)

Now we will put the correct discount factor in A11 by replacing its entry with this formula:

= 1 / $A$10

Our final step is to put the formula for the arbitrage pricing probability, q, in A12. This probability is given in Chapter 2; see equation (2.5). The formula is

= ($A$10-$A$7) / ($A$6-$A$7)

Once you have entered this formula, any changes in the parameters u, d, r, the stock price, or the strike price will produce the correct tree. Think about this last statement. Your spreadsheet, once constructed, will carry out tedious computation quickly and effortlessly.

EXERCISES

1. Use a "Euro Call" spreadsheet to find the call price given that the stock price at $t = 0$ is \$120, the exercise price is \$130, expiration is 5 periods into the future, the interest rate per period is 4.5%, $u = 1.05$, and $d = 0.88$. Record the option's price at the bottom of your spreadsheet. Then change the exercise price to \$125 and to \$120, and record the new option prices on your spreadsheet. Print your spreadsheet.

2. Use a "Euro Call" spreadsheet to find the call price given that the stock price at $t =$ 0 is \$65, the exercise price is \$70, expiration is 5 periods into the future, the interest rate per period is 5%, $u = 1.1$, and $d = 0.91$. Record the option's price at the bottom of your spreadsheet. Then change the interest rate to 6% and to 7%, and record the new option prices on your spreadsheet. Print your spreadsheet. Why does the price increase as r increases?
3. Use the "Euro Call" spreadsheet of Exercise 1 to find the call price given the same stock parameters and $X = 130$. However, the expiration is **8** periods into the future.

Hint: You must copy and paste lines 17 and 27 to produce 3 new time periods.

4. Use the "Euro Call" spreadsheet of Exercise 2 to find the call price given the same stock parameters and $X = 70$. However, the expiration is **7** periods into the future.

Hint: You must copy and paste lines 17 and 27 to produce new time periods.

4.3 COMPUTING AMERICAN OPTION TREES

Sometimes it is profitable to exercise an option before its expiration date. As stated in Section 1.2.2, an option that allows this possibility is called an *American option.* The option's price can be computed with the tree method, but, as was explained in Section 3.3, the node entries must be tested in a certain order so that the replicating-portfolio method applies to the tree.

We have an option tree spreadsheet, created in Section 4.2, that performs the chaining steps for a European call. It is easy to adjust its formulas to check for early exercise, but no new price emerges from this adjustment. There is a financial reason why early exercise for calls is unprofitable.[1] However, American puts are more interesting.

We will modify the option tree from Section 4.2 so that it prices **American puts**. Use line 16 of the spreadsheet to record the possible expiration prices for a put that expires after five periods.

What are these option values? They are 0 if the stock price is higher than the strike price, and they are the difference of the strike and the stock price if the stock is lower than the strike. To state this briefly, we may say that the value is 0 unless "strike" minus "stock" is positive and then we take the latter amount. The appropriate spreadsheet formula that expresses this rule, for cell K16, is

= MAX($A$2-K6 , 0)

Recall that A2 is involved because that is where the strike price is stored. Enter this formula and **save** this spreadsheet, giving it a different file name such as "Amer Put." The formula you just entered refers to A2 to obtain the strike and subtracts the number in K6 from it. If the difference is negative, it records a 0 in the cell.

---

[1] Hull, J. C., *Options, Futures, and Other Derivatives,* 3d ed., Prentice Hall, Upper Saddle River, NJ. 1997, pp. 162–165.